Rules for integrands of the form $(d \operatorname{Sin}[e+f x])^m (a+b \operatorname{Tan}[e+f x])^n$

1: $\int \operatorname{Sin}[e+f x]^m (a+b \operatorname{Tan}[e+f x])^n \, dx$ when $\frac{m}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: $\operatorname{Sin}[e+f x]^2 == \frac{\operatorname{Tan}[e+f x]^2}{1+\operatorname{Tan}[e+f x]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then

$\operatorname{Sin}[e+f x]^m F[b \operatorname{Tan}[e+f x]] == \frac{b}{f} \operatorname{Subst}\left[\frac{x^m F[x]}{\left(b^2+x^2\right)^{\frac{m}{2}+1}}, x, b \operatorname{Tan}[e+f x]\right] \partial_x (b \operatorname{Tan}[e+f x])$

Rule: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\int \operatorname{Sin}[e+f x]^m (a+b \operatorname{Tan}[e+f x])^n \, dx \rightarrow \frac{b}{f} \operatorname{Subst}\left[\int \frac{x^m (a+x)^n}{\left(b^2+x^2\right)^{\frac{m}{2}+1}} \, dx, x, b \operatorname{Tan}[e+f x]\right]$$

–

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*tan[e_.+f_.*x_])^n_,x_Symbol] :=
  b/f*Subst[Int[x^m*(a+x)^n/(b^2+x^2)^(m/2+1),x],x,b*Tan[e+f*x]] /;
FreeQ[{a,b,e,f,n},x] && IntegerQ[m/2]
```

2. $\int \sin[e+f x]^m (a+b \tan[e+f x])^n \, dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

1: $\int \sin[e+f x]^m (a+b \tan[e+f x])^n \, dx$ when $\frac{m-1}{2} \in \mathbb{Z} \land n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \sin[e+f x]^m (a+b \tan[e+f x])^n \, dx \longrightarrow \int \text{Expand}\left[\sin[e+f x]^m (a+b \tan[e+f x])^n, x\right] dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Expand[Sin[e+f*x]^m*(a+b*Tan[e+f*x])^n,x],x] /;
FreeQ[{a,b,e,f},x] && IntegerQ[(m-1)/2] && IGtQ[n,0]
```

2: $\int \sin[e+f x]^m (a+b\tan[e+f x])^n \, dx$ when $\frac{m-1}{2} \in \mathbb{Z} \land n \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $a+b\tan[z] = \frac{a\cos[z]+b\sin[z]}{\cos[z]}$

Note: This rule sucks...

Rule: If $\frac{m-1}{2} \in \mathbb{Z} \land n \in \mathbb{Z}^-$, then

$$\int \sin[e+f x]^m (a+b\tan[e+f x])^n \, dx \to \int \frac{\sin[e+f x]^m (a\cos[e+f x]+b\sin[e+f x])^n}{\cos[e+f x]^n} \, dx$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Sin[e+f*x]^m*(a*Cos[e+f*x]+b*Sin[e+f*x])^n/Cos[e+f*x]^n,x] /;
FreeQ[{a,b,e,f},x] && IntegerQ[(m-1)/2] && ILtQ[n,0] && (LtQ[m,5] && GtQ[n,-4] || EqQ[m,5] && EqQ[n,-1])
```

Rules for integrands of the form $(d\csc[e+f x])^m (a+b\tan[e+f x])^n$

1: $\int (d\csc[e+f x])^m (a+b\tan[e+f x])^n \, dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((d\csc[e+f x])^m \left(\frac{\sin[e+f x]}{d}\right)^m \right) = 0$

Rule: If $m \notin \mathbb{Z}$, then

$$\int (d\,\mathrm{Csc}[e+f x])^m\,(a+b\,\mathrm{Tan}[e+f x])^n\,dx \,\longrightarrow\, (d\,\mathrm{Csc}[e+f x])^{\mathrm{FracPart}[m]}\left(\frac{\mathrm{Sin}[e+f x]}{d}\right)^{\mathrm{FracPart}[m]}\int\frac{(a+b\,\mathrm{Tan}[e+f x])^n}{\left(\frac{\mathrm{Sin}[e+f x]}{d}\right)^m}\,dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^m_*(a_.+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  (d*Csc[e+f*x])^FracPart[m]*(Sin[e+f*x]/d)^FracPart[m]*Int[(a+b*Tan[e+f*x])^n/(Sin[e+f*x]/d)^m,x] /;
FreeQ[{a,b,d,e,f,m,n},x] && Not[IntegerQ[m]]
```

Rules for integrands of the form $\mathrm{Cos}[e+f x]^m\,\mathrm{Sin}[e+f x]^p\,(a+b\,\mathrm{Tan}[e+f x])^n$

1: $\int \mathrm{Cos}[e+f x]^m\,\mathrm{Sin}[e+f x]^p\,(a+b\,\mathrm{Tan}[e+f x])^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: $a+b\,\mathrm{Tan}[z] == \frac{a\,\mathrm{Cos}[z]+b\,\mathrm{Sin}[z]}{\mathrm{Cos}[z]}$

Rule: If $n\in\mathbb{Z}$, then

$$\int \mathrm{Cos}[e+f x]^m\,\mathrm{Sin}[e+f x]^p\,(a+b\,\mathrm{Tan}[e+f x])^n\,dx \,\longrightarrow\, \int \mathrm{Cos}[e+f x]^{m-n}\,\mathrm{Sin}[e+f x]^p\,(a\,\mathrm{Cos}[e+f x]+b\,\mathrm{Sin}[e+f x])^n\,dx$$

Program code:

```
Int[cos[e_.+f_.*x_]^m_.*sin[e_.+f_.*x_]^p_.*(a_+b_.*tan[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Cos[e+f*x]^(m-n)*Sin[e+f*x]^p*(a*Cos[e+f*x]+b*Sin[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,m,p},x] && IntegerQ[n]
```

```
Int[sin[e_.+f_.*x_]^m_.*cos[e_.+f_.*x_]^p_.*(a_+b_.*cot[e_.+f_.*x_])^n_.,x_Symbol] :=
  Int[Sin[e+f*x]^(m-n)*Cos[e+f*x]^p*(a*Sin[e+f*x]+b*Cos[e+f*x])^n,x] /;
FreeQ[{a,b,e,f,m,p},x] && IntegerQ[n]
```